```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

We consider a partial differential system studied in J.-F. Pommaret, chapter V of the book *Advanced Topics in Control Systems Theory*, Lecture Notes in Control and Information Sciences 311, F. Lamnabhi-Lagarrigue, A. Loria, E. Panteley Editors, Springer, 2005, 155-223, and defined by the following matrix:

```
> A:=DefineOreAlgebra(diff=[d[1],x[1]],diff=[d[2],x[2]],polynom=[x[1],x[2]],comm=[a]):
> R:=matrix(3,3,[0,d[2]-d[1],d[2]-d[1]-a,d[2],-d[1],-d[2]-d[1]-a,d[1],-d[1],-2*d[1]]);
```

$$R := \begin{bmatrix} 0 & d_2 - d_1 & d_2 - d_1 - a \\ d_2 & -d_1 & -d_2 - d_1 - a \\ d_1 & -d_1 & -2\,d_1 \end{bmatrix}$$

Let us consider the ring $A = \mathbb{Q}(a)[d_1, d_2]$ and $M = A^{1\times 3}/(A^{1\times 3}\,R)$ the A-module finitely presented by the matrix R. We denote by π the projection from $A^{1\times 3}$ to M. We can compute the A-module structure of the endomorphism ring $E = \mathrm{end}_A(M)$ of M:

```
> Endo:=MorphismsConstCoeff(R,R,A):
> Endo[1];
```

$$[\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 1 & -1 & -2 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & d_1 - d_2 - a \\ 0 & 0 & d_2 - d_1 - a \\ 0 & 0 & -d_2 + d_1 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 2 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}]$$

Hence, we obtain that E is finitely generated over A by 4 elements $f_1 = \mathrm{id}_M$, f_2, f_3, f_4 defined by $f_i(m) = \pi(\lambda\, Endo[1, i])$, where $m = \pi(\lambda) \in M$, $\lambda \in A^{1\times 3}$ and $i = 1, \ldots, 4$. These generators satisfy the A-linear relations $Endo[2]\,(f_1 \quad f_2 \quad f_3 \quad f_4)^T = 0$, where $Endo[2]$ is defined by:

```
> Endo[2];
```

$$\begin{bmatrix} d_2 & 0 & 0 & -d_2 \\ d_1 & 0 & -1 & -d_2 \\ 0 & d_2 & 0 & 0 \\ 0 & d_1 & 0 & 0 \\ 0 & 0 & -1 & -d_2 + d_1 \end{bmatrix}$$

Let us try to find idempotent elements of E defined by means of idempotent matrices $P \in \mathbb{Q}(a)^{3\times 3}$ and $Q \in \mathbb{Q}(a)^{3\times 3}$, namely, $e \in E$ satisfying $e^2 = e$, where $e(\pi(\lambda)) = \pi(\lambda\, P)$, for all $\lambda \in A^{1\times 3}$, and $R\,P = Q\,R$, $P^2 = P$, $Q^2 = Q$:

```
> Idem:=IdempotentsMatConstCoeff(R,Endo[1],A,0,alpha):Idem[1];
```

$$[\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} -c41 & c41 & 2\,c41 \\ -c41 - 1 & 1 + c41 & 2\,c41 + 2 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 - c41 & c41 & 2\,c41 \\ 1 - c41 & c41 & -2 + 2\,c41 \\ 0 & 0 & 1 \end{bmatrix}]$$

We obtain two non-trivial idempotent endomorphisms e_1 and e_2 of E respectively defined by the matrices $Idem[1, 3]$ and $Idem[1, 4]$. Let us consider e_1 defined by the following matrices $P = Idem[1, 3] \in A^{3\times 3}$ where we have set to 0 the arbitrary constant $c41$ and $Q \in A^{3\times 3}$ satisfying $R\,P = Q\,R$:

```
> P:=subs(c41=0,Idem[1,3]); Q:=Factorize(Mult(R,P,A),R,A);
```

$$P := \begin{bmatrix} 0 & 0 & 0 \\ -1 & 1 & 2 \\ 0 & 0 & 0 \end{bmatrix} \quad Q := \begin{bmatrix} 1 & -1 & 1 \\ 0 & 0 & 1 \\ 0 & 0 & 1 \end{bmatrix}$$

As we have $P^2 = P$ and $Q^2 = Q$, we know that the A-modules $\ker_A(.P)$, $\mathrm{im}_A(.P) = \ker_A(.(I_3 - P))$, $\ker_A(.Q)$ and $\mathrm{im}_A(.Q) = \ker_A(.(I_3 - Q))$ are projective, and thus, free by the Quillen-Suslin theorem. Let us compute bases of the corresponding modules. As the coefficients of P and Q belong to $\mathbb{Q}$, we can obtain them by means of linear algebraic techniques (e.g., using the *jordan* command of Maple) or using directly OreModules as it is explained below. We then form the matrices U and V such that $U\,P\,U^{-1}$ and $V\,Q\,V^{-1}$ are the Jordan normal forms of P and Q.

```
> U1:=SyzygyModule(P,A): U2:=SyzygyModule(evalm(1-P),A):
> U:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q,A): V2:=SyzygyModule(evalm(1-Q),A):
> V:=stackmatrix(V1,V2);
```

$$U := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 0 & 1 \\ 1 & -1 & -2 \end{bmatrix} \quad V := \begin{bmatrix} 0 & 1 & -1 \\ 1 & -1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

We obtain that the two unimodular matrices U and V, i.e., $U \in \mathrm{GL}_3(A)$ and $V \in \mathrm{GL}_3(A)$, satisfy that the matrix $V\,R\,U^{-1}$ is block-diagonal:

```
> S:=Mult(V,R,LeftInverse(U,A),A);
```

$$S := \begin{bmatrix} d_2 - d_1 & d_1 - d_2 - a & 0 \\ 0 & 0 & -d_2 \\ 0 & 0 & d_1 \end{bmatrix}$$

We can also use the command *HeuristicDecomposition* to directly obtain the previous result:

```
> HeuristicDecomposition(R,P,A)[1];
```

$$\begin{bmatrix} d_2 - d_1 & d_1 - d_2 - a & 0 \\ 0 & 0 & -d_2 \\ 0 & 0 & d_1 \end{bmatrix}$$

We obtain that the general solution $y = (y_1(x_1,x_2) \quad y_2(x_1,x_2) \quad y_3(x_1,x_2))^T$ of the system $R\,y = 0$ is given by $y = U^{-1}\,z$, where $z = (z_1(x_1,x_2) \quad z_2(x_1,x_2) \quad z_3(x_1,x_2))^T$ is the general solution of $S\,z = 0$. Hence, if we denote by $\mathcal{F} = C^\infty(\mathbb{R}^2)$, $\ker_{\mathcal{F}}(R.) = \{y \in \mathcal{F}^3 \mid R\,y = 0\}$ and $\ker_{\mathcal{F}}(S.) = \{z \in \mathcal{F}^3 \mid S\,z = 0\}$, then we have $\ker_{\mathcal{F}}(R.) = U^{-1}\,\ker_{\mathcal{F}}(S.)$.

If $a \neq 0$, then $\ker_{\mathcal{F}}(S.) = \{z \in \mathcal{F}^3 \mid S\,z = 0\}$ is parametrized by

$$\begin{cases} z_1(x_1,x_2) = (d_2 - d_1 + a)\,u(x_1,x_2), \\ z_2(x_1,x_2) = (d_2 - d_1)\,u(x_1,x_2), \\ z_3(x_1,x_2) = c, \end{cases}$$

where u is an arbitrary function of $\mathcal{F}$ and c an arbitrary constant. For more details, see F. Chyzak, A. Quadrat, D. Robertz, "Effective algorithms for parametrizing linear control systems over Ore algebras", *Appl. Algebra Engrg. Comm. Comput.*, 16 (2005), 319-376. This last result can be checked as follows:

```
> L[1]:=Parametrization(S,A);
```

$$L_1 := \begin{bmatrix} a\,\xi_1(x_1,x_2) + \frac{\partial}{\partial x_2}\xi_1(x_1,x_2) - \frac{\partial}{\partial x_1}\xi_1(x_1,x_2) \\ \frac{\partial}{\partial x_2}\xi_1(x_1,x_2) - \frac{\partial}{\partial x_1}\xi_1(x_1,x_2) \\ \_C1 \end{bmatrix}$$

Using the fact that the matrices U and V do not depend on a and

```
> U_inv:=inverse(U);
```

$$U\_inv := \begin{bmatrix} 1 & 0 & 0 \\ 1 & -2 & -1 \\ 0 & 1 & 0 \end{bmatrix}$$

we obtain the following parametrization of $\ker_{\mathcal{F}}(R.) = \{y \in \mathcal{F}^3 \mid R\, y = 0\}$

```
> Sol[1]:=ApplyMatrix(U_inv,L[1],A);
```

$$Sol_1 := \begin{bmatrix} a\,\xi_1(x_1,x_2) + \frac{\partial}{\partial x_2}\xi_1(x_1,x_2) - \frac{\partial}{\partial x_1}\xi_1(x_1,x_2) \\ a\,\xi_1(x_1,x_2) - \frac{\partial}{\partial x_2}\xi_1(x_1,x_2) + \frac{\partial}{\partial x_1}\xi_1(x_1,x_2) - \_C1 \\ \frac{\partial}{\partial x_2}\xi_1(x_1,x_2) - \frac{\partial}{\partial x_1}\xi_1(x_1,x_2) \end{bmatrix}$$

i.e., all the $\mathcal{F}$-solutions of the system $R\,y = 0$ have the form:

$$\forall\, u \in \mathcal{F}, \quad \forall\, c \in \mathbb{R}, \quad \begin{cases} y_1(x_1,x_2) = (d_2 - d_1 + a)\, u(x_1,x_2), \\ y_2(x_1,x_2) = (d_1 - d_2 + a)\, u(x_1,x_2) - c, \\ y_3(x_1,x_2) = (d_2 - d_1)\, u(x_1,x_2). \end{cases}$$

If $a = 0$, using Theorem 6 of A. Quadrat, D. Robertz, "Parametrizing all solutions of uncontrollable multidimensional linear systems", *Proceedings of* 16^{th} *IFAC World Congress*, Prague (Czech Republic), 04-08/07/05, we then obtain that the $\mathcal{F}$-solutions of $(d_2 - d_1)\, z_1 + (d_1 - d_2)\, z_2 = 0$ are parametrized by:

$$\forall\, u \in \mathcal{F}, \quad \forall\, \phi \in C^\infty(\mathbb{R}), \quad \begin{cases} z_1(x_1,x_2) = u(x_1,x_2), \\ z_2(x_1,x_2) = u(x_1,x_2) - \phi(x_1 + x_2). \end{cases}$$

Hence, we get the following parametrization of $\ker_{\mathcal{F}}(S.)$:

$$\forall\, u \in \mathcal{F}, \quad \forall\, \phi \in C^\infty(\mathbb{R}), \quad c \in \mathbb{R}, \quad \begin{cases} z_1(x_1,x_2) = u(x_1,x_2), \\ z_2(x_1,x_2) = u(x_1,x_2) - \phi(x_1 + x_2), \\ z_3(x_1,x_2) = c. \end{cases}$$

We can check this result as follows:

```
> T:=subs(a=0,evalm(S));
```

$$T := \begin{bmatrix} d_2 - d_1 & d_1 - d_2 & 0 \\ 0 & 0 & -d_2 \\ 0 & 0 & d_1 \end{bmatrix}$$

```
> L[2]:=Parametrization(T,A);
```

$$L_2 := \begin{bmatrix} \xi_1(x_1,x_2) \\ -\_F1(x_2 + x_1) + \xi_1(x_1,x_2) \\ \_C1 \end{bmatrix}$$

Then, $\ker_{\mathcal{F}}(R.) = U^{-1}\, \ker_{\mathcal{F}}(S.)$ is parametrized by

```
> Sol[2]:=ApplyMatrix(U_inv,L[2],A);
```

$$Sol_2 := \begin{bmatrix} \xi_1(x_1,x_2) \\ -\xi_1(x_1,x_2) + 2\,\_F1(x_2 + x_1) - \_C1 \\ -\_F1(x_2 + x_1) + \xi_1(x_1,x_2) \end{bmatrix}$$

i.e., we have the following parametrization of $\ker_{\mathcal{F}}(R.)$:

$$\forall\, u \in \mathcal{F}, \quad \forall\, \phi \in C^{\infty}(\mathbb{R}), \quad c \in \mathbb{R}, \quad \left\{ \begin{array}{l} y_1(x_1, x_2) = u(x_1, x_2), \\ y_2(x_1, x_2) = 2\,\phi(x_1 + x_2) - u(x_1, x_2) - c, \\ y_3(x_1, x_2) = -\phi(x_1 + x_2) + u(x_1, x_2). \end{array} \right.$$

To conclude, if $a \neq 0$, then we obtain that $\ker_{\mathcal{F}}(R.)$ depends on one arbitrary function of two independent variables and one arbitrary constant, whereas if $a = 0$, $\ker_{\mathcal{F}}(R.)$ depends on one arbitrary function of two independent variables, one arbitrary constant but also on one arbitrary function of one independent variable. This result proves again in an unified way the results obtained in J.-F. Pommaret, chapter V of the book *Advanced Topics in Control Systems Theory*, Lecture Notes in Control and Information Sciences 311, F. Lamnabhi-Lagarrigue, A. Loria, E. Panteley Editors, Springer, 2005, 155-223.